A study of the variation of the focusing term `M21` and coupling term `M41` of the transformation corresponding to a `beambeam` element as a function of its `xma` and `yma` attributes was performed. `xma` and `yma` are the horizontal and vertical displacements of the opposite beam with respect to the ideal orbit. Values ranging from 10^{-2} m to 10^{-18} m with steps of one order of magnitude were assigned to both. A scan for each coordinate while keeping a zero-displacement in the opposite coordinate was also included. In order to get the `M21` and `M41` terms, the sector map of a sequence consisting of a single `beambeam` element was obtained with MAD-X. The initial conditions of the `twiss` computation, and other fixed atributes of the `beambeam` element are listed in the source code below.

```
option, echo, info, warn;

twissmacro(ii, jj): macro = {
   use, sequence = bbseq;
   select, flag = twiss, clear;
   twiss, betx = 2.99098698, bety = 2.940854601, x = 0.0, y = 0.0, sectormap;
}

on_ho2 = 1.0;

beam, particle = proton, bunched = true, radiate = false, energy:= 7000,
      ex:= 5.026457389e-10, exn := 1.499999987e-05, ey := 5.026457389e-10,
      eyn := 1.499999987e-05, et := 1, sigt := 0.0755, sige := 0.00011,
      npart := 1.3e+11, bcurrent:= 0.000234224583, freq0 := 0.01124549941,
      circ := 26658.8832, dtbyds := 0, deltap := 0, alfa := 1.796641699e-08,
      u0 := 0, qs := 0, arad := 1.534698269e-18, pdamp := {1,1,2}, n1min := -1;

nsteps  = 18;
fstep   = 10;
stepmin = 1e-18;

ii  = 0;
xii = 0.0;
create, table = mytable, column = ii, xii, jj, yjj, re21, re41;
while (ii < nsteps) {

  jj  = 0;
  yjj = 0.0;
  while (jj < nsteps) {

    bb: beambeam, sigx := 3.883216731e-05, sigy := 3.883216724e-05,
                  xma:= xii, yma:= yjj, charge :=  on_ho2;

    bbseq: sequence, l = 1e-9;
    bb, at = 0.0;
    endsequence;

    exec, twissmacro($ii, $jj);
    re21 = table(sectortable, bb, r21);
    re41 = table(sectortable, bb, r41);
    fill, table = mytable;

    if (jj == 0) { yjj = stepmin; }
    else         { yjj = yjj*fstep; }

    jj = jj + 1;

  }

  if (ii == 0) { xii = stepmin; }
```

```
  else         { xii = xii*fstep; }

  ii = ii + 1;
}

write, table = mytable, file = "mytable.tfs";
```

Fig. 1 (left) shows the value of the focusing term `M21` of the transport matrix `R` at the `beambeam` element. An abrupt change in the corner of the window `xma` $\leq 10^{-13}$ m, `yma` $\leq 10^{-13}$ m is observed. This contradicts the expected behaviour of a constant value in the area around (O,O), the point without horizontal and vertical displacements. In Fig. 1 (right) an induced coupling appears for displacements in the diagonal (`xma` = `yma`) of the same window.

The effect on the beta-beating is observed in the LHC with `beambeam` elements added to the sequence around IP1 (head-on), as seen in Fig. 2. A negligible offset has been added in order to avoid the numerical problems. Nevertheless, this changes the beta-beating in both planes slightly.



Figure 1: Focusing and coupling terms of the transformation matrix at the `beambeam` element.



Figure 2: Beta-beating in the LHC due to beam-beam at the IP1. The cases with and without separation correspond to `on_sep1` = `on_sep1v` = 10^{-6} and `on_sep1` = `on_sep1v` = 0.0 respectively.